
Lab 5: Covariance Modeling

Overview

1. Random Effects Modeling.

a. Random intercept (RI)

b. Random intercept and slope (RIAS)

c. Random intercept and slope and quadratic (RIASAQ)
2. Repeated Statement

a. SIMPLE

b. CS

c. AR(1)

d. ARMA(1,1)

e. ANTE(1)

f. UN

g. UNR

h. CSH

i. ARH(1)

j. FA(1)

k. FA(2)

l. FA1(1)

m. FA1(2)

n. AR+RI

3. Macro to fit a number of models at once and make a nice table.  

Small Mice Data Empirical Variances and Correlations

	Vars
	886
	2272
	3400
	10513
	17884
	18835
	14560

	Corrs
	 
	 
	 
	 
	 
	 
	 

	Day
	   2
	   5
	   8
	   11
	   14
	   17
	   20

	 2
	1.
	.92
	.57
	.36
	.23
	.23
	.38

	 5
	.92
	1.
	.77
	.54
	.45
	.41
	.55

	 8
	.57
	.77
	1.
	.86
	.80
	.76
	.81

	11
	.36
	.54
	.86
	1.
	.93
	.92
	.87

	14
	.23
	.45
	.80
	.93
	1.
	.96
	.89

	17
	.23
	.41
	.76
	.92
	.96
	1.
	.92

	20
	.38
	.55
	.81
	.87
	.89
	.92
	1.


These are for comparing to the output from our SAS runs.  

Lab Tasks

0. Use the small mice data set to fit various covariance models.  

data smallmice;

input Group id w1 w2 w3 w4 w5 w6 w7;

weight=w1; day=2;  cday=day; output;

weight=w2; day=5;  cday=day; output;

weight=w3; day=8;  cday=day; output;

weight=w4; day=11; cday=day; output;

weight=w5; day=14; cday=day; output;

weight=w6; day=17; cday=day; output;

weight=w7; day=20; cday=day; output;

drop w1 w2 w3 w4 w5 w6 w7;

cards;

       3      22    190    388    621    823    1078   1132   1191

       3      23    218    393    568    729    839    852    1004

       3      24    141    260    472    662    760    885    878

       3      25    211    394    549    700    783    870    925

       3      26    209    419    645    850    1001   1026   1069

       3      27    193    362    520    530    641    640    751

       3      28    201    361    502    530    657    762    888

       3      29    202    370    498    650    795    858    910

       3      30    190    350    510    666    819    879    929

       3      31    219    399    578    699    709    822    953

       3      32    225    400    545    690    796    825    836

       3      33    224    381    577    756    869    929    999

       4      34    187    329    441    525    589    621    796

       4      35    278    471    606    770    888    1001   1105

;

run;

proc sort;

by id day;

run;

proc print;

run;

1. Random Statement

Run three random effects models, RI, RIAS, RIASAQ on the small mice data.  We will use the same quadratic fixed effects model in all of our analyses.  Random effects models can be applied to data with random times, as well as balanced and balanced with missing data.  

title 'RI';

proc mixed method=reml covtest noclprint;

class id;

model weight=day day*day / s;

random intercept / type=un subject=id v vcorr;

run;

title 'RIAS';

proc mixed method=reml covtest noclprint;

class id;

model weight=day day*day / s;

random intercept day / type=un subject=id v vcorr;

run;

title 'RIASAQ';

proc mixed method=reml covtest noclprint;

class id;

model weight=day day*day / s;

random intercept day day*day / type=un subject=id v vcorr;

run;

Keywords:


method=reml, use reml likelihood estimation.  This is the default.  


covtest ~ se, t, p for the covariance parameters


noclprint ~ don’t print some basic info about the data set.  After your 
 

models are running satisfactorily, put this in.  Omit to start.  


s
~ give table of parameter estimates, SE’s, t, p


type = un ~ Use the unstructured covariance matrix for the random 
  

effects covariance matrix D.  This is usually what we want to 
 

do for random effects. 


subject=id ~ Naming the id variable to identify that observations are 
 

from a single individual.  


v 
~ give the fitted covariance matrix for the first subject


vcorr 
~ give the fitted correlation matrix for the first subject

TODO: (0)  Check that the algorithm has converged.  Look in the log, and also below ‘iteration history’.  

TODO: (1)  Record the fit statistics and the residual variance in the table located at the end of this handout.  

TODO: (2)  Find the covariance parameter estimates for each model, identify the names of each parameter.  In these random effects models, there is the 
Variance of the random intercept.
Variance of the random slope (if present),
Variance of the random quadratic (if present), and
covariances (0 1 or 3 for the three models) between those random effects present, and the
residual variance 2.  
TODO:  (3a) Compare the fitted correlations to the raw correlation matrix given on page 1.  

TODO:  (3b) Compare the fitted variances to the raw variances given on page 1.  

TODO:  (4) Do any of these three models come remotely close to fitting the observed correlations?  

2. Repeated Statement

The repeated statement is used to tell SAS to use all the other (non-random effects) covariance models.  Models accessed through the repeated statement usually need to be applied to data sets with balanced (with missing) data sets only.  

a. Putting in time in the repeated statement.  

The following two sets of SAS codes fit the AR(1) in two ways.  The variable ‘cday’ is the same as ‘day’.  We need two copies of the same variable because we wish to use one in the model statement and one in the repeated statement.  The first repeated statement assumes balanced, or balanced with missing data: the data set needs to be organized so that each subject has the same number of observations and therefore every subject has the same number (7 in this caes) of lines in the data set.  The second version is more general in that the data set does not need to have a row for observations that are missing the outcome.  

title 'AR(1) no cday';

proc mixed method=reml covtest noclprint;

class id;

model weight=day day*day / s;

repeated / type=ar(1) subject=id   r rcorr;

run;

title 'AR(1)';

proc mixed method=reml covtest noclprint;

class id cday;

model weight=day day*day / s;

repeated cday / type=ar(1) subject=id   r rcorr;

run;

You’ll get the same results both times, because the small mice data is balanced without any missing data.  In larger more complicated data sets, where missing observations are common, we have to either (a) fit the data using the second version of the code or (b) construct a data set with rows for all the missing outcome values.  

Continue the lab with the second version of the code.  

The keyword r requests SAS print the covariance matrix for the first subject.  Changing this to r=3 or r=3,5,12 would ask for the correlation matrix for the 3rd, or the 3rd, 5th and 12th, subjects respectively.  Keyword rcorr asks for the correlation matrix and rcorr=3 or rcorr=3,5,12 asks for the correlation matrix for the 3rd, or several subjects. 
b. Trying out a number of covariance models.  

TODO:  Alter the second AR(1) SAS code example to fit all of these covariance models in turn.  Use the results to fill in the table following on page 6.  For each model, you need to change the title and the repeated statement:
repeated cday / type=simple subject=id   r rcorr;

repeated cday / type=cs subject=id   r rcorr;

repeated cday / type=ar(1) subject=id   r rcorr;

repeated cday / type=arma(1,1) subject=id   r rcorr;

repeated cday / type=ante(1) subject=id   r rcorr;

repeated cday / type=un subject=id   r rcorr;

repeated cday / type=unr subject=id   r rcorr;

repeated cday / type=csh subject=id   r rcorr;

repeated cday / type=arh(1) subject=id   r rcorr;

repeated cday / type=fa(1) subject=id   r rcorr;

repeated cday / type=fa(2) subject=id   r rcorr;

repeated cday / type=fa1(1) subject=id   r rcorr;

repeated cday / type=fa1(2) subject=id   r rcorr;

As with the random effects models, you should do the same basic steps:  

TODO: (0)  Check that the algorithm has converged.  Look in the SAS log, and also below ‘iteration history’.  

TODO: (1)  Record the fit statistics and the residual variance in the table located at the end of this handout.  

TODO: (2)  Find the covariance parameter estimates for each model, identify the names of each covariance parameter.  Some of this may need to wait till we discuss these models in class!  
TODO:  (3a) Compare the fitted correlations to the raw correlation matrix given on page 1.  

TODO:  (3b) Compare the fitted variances to the raw variances given on page 1.  

TODO:  (4) Which model(s) produce variance and correlation fit the observed correlations and variances?  

TODO:  (5)  Compare the RI and CS output.  Confirm that these are the same model except for the df.  

3. Repeated and Random Statement Together

There can be at most one repeated statement in a model.  There can be more than one random statement and there can be both a repeated and a random statement in the same model.  Here are two examples.  The first runs on the small mice data.  The second is a very popular covariance model that competes with the ARMA(1,1) model, although they are not the same.  The second model does not converge properly on my computer and gives both an ok and an error message, 

NOTE: Convergence criteria met.

NOTE: Estimated G matrix is not positive definite.

Inspection of the output shows that the UN(1,1) parameter has been forced equal to zero, and that the fit is equivalent to the AR(1) model fit.  In other models, this may not happen; the mice data have virtually no random intercept component, and it is not needed in trying to fit this model to this data set.  In general, don’t trust the output when you get this error message.  
title 'ARH(1) + RI';

proc mixed method=reml covtest noclprint;

class id cday;

model weight=day day*day / s;

random intercept /type=un subject=id v vcorr;

repeated cday / type=arh(1) subject=id   r;

run;

title 'AR(1) + RI';

proc mixed method=reml covtest noclprint;

class id cday;

model weight=day day*day / s;

random intercept /type=un subject=id v vcorr;

repeated cday / type=ar(1) subject=id   r;

run;

The output for the second set of code is nearly equal to the code from the model without the random statement.  Compare without that statement to the output with that code.  Look at the fixed effects output in particular.  

TO DO TASKS

Fill out the table as you work.

	Random
	#Parms
	REML
	AIC
	BIC
	2

	
	RI
	
	
	
	
	 

	
	RIAS
	
	
	
	
	 

	
	RIASAQ
	 
	 
	 
	 
	 

	Repeated
	
	
	
	
	 

	
	SIMPLE
	
	
	
	
	 

	
	CS
	
	
	
	
	 

	
	AR(1)
	
	
	
	
	 

	
	ARMA(1,1)
	
	
	
	
	 

	
	ANTE(1)
	 
	 
	 
	 
	 

	
	UN
	
	
	
	
	 

	
	UNR
	
	
	
	
	 

	
	CSH
	
	
	
	
	 

	
	ARH(1)
	 
	 
	 
	 
	 

	
	FA(1)
	
	
	
	
	 

	
	FA(2)
	
	
	
	
	 

	
	FA1(1)
	
	
	
	
	 

	
	FA1(2)
	 
	 
	 
	 
	 

	
	AR+RI
	 
	 
	 
	 
	 


4. Macro to run all the models at once and create a table of fitted values.  
(Courtesy Leanne Streja)  
*CODE TO MAKE TABLE FOR LAB5

LEANNE STREJA

FEB 7, 2005;
data smallmice;

input Group id w1 w2 w3 w4 w5 w6 w7;

weight=w1; day=2;  cday=day; output;

weight=w2; day=5;  cday=day; output;

weight=w3; day=8;  cday=day; output;

weight=w4; day=11; cday=day; output;

weight=w5; day=14; cday=day; output;

weight=w6; day=17; cday=day; output;

weight=w7; day=20; cday=day; output;

drop w1 w2 w3 w4 w5 w6 w7;

cards;

       3      22    190    388    621    823    1078   1132   1191

       3      23    218    393    568    729    839    852    1004

       3      24    141    260    472    662    760    885    878

       3      25    211    394    549    700    783    870    925

       3      26    209    419    645    850    1001   1026   1069

       3      27    193    362    520    530    641    640    751

       3      28    201    361    502    530    657    762    888

       3      29    202    370    498    650    795    858    910

       3      30    190    350    510    666    819    879    929

       3      31    219    399    578    699    709    822    953

       3      32    225    400    545    690    796    825    836

       3      33    224    381    577    756    869    929    999

       4      34    187    329    441    525    589    621    796

       4      35    278    471    606    770    888    1001   1105

;

run;

proc sort;

by id day;

run;

/*proc print;

run;*/
%macro model(i,title,rr,what,type);

title "&title";

proc mixed method=reml covtest noclprint data=smallmice;

class id cday;

model weight=day day*day / s;

&rr &what / type=&type subject=id;

ods output covparms=c&i fitstatistics=f&i dimensions=d&i;

run;

data new&i (keep=descr value type model);

format descr $30.;

set c&i(rename=(covparm=descr estimate=value)) f&i d&i;

if (index(descr,'Res')>0 or index(descr, 'm')>0) and index(descr,'CC')=0 and index(descr,'mm')=0;

type="&title";

model=&i;

run;

proc sort data=new&i;by descr;run;

%mend;

%model(1,RI, random, intercept,un);

%model(2,RIAS, random, intercept day,un);

%model(3,RIASAQ, random, intercept day day*day,un);

%model(4,SIMPLE,repeated, cday,SIMPLE);

%model(5,CS,repeated, cday,CS);

%model(6,AR(1),repeated, cday,AR(1));

%model(7,ARMA(1,1),repeated, cday,ARMA(1,1));

%model(8,ANTE(1),repeated, cday,ANTE(1));

%model(9,UN,repeated, cday,UN);

%model(10,UNR,repeated, cday,UNR);

%model(11,CSH,repeated, cday,CSH);

%model(12,ARH(1),repeated, cday,ARH(1));

%model(13,FA(1),repeated, cday,FA(1));

%model(14,FA(2),repeated, cday,FA(2));

%model(15,FA1(1),repeated, cday,FA1(1));

%model(16,FA1(2),repeated, cday,FA1(2));

*Model 17;
title 'AR(1) + RI';

proc mixed method=reml covtest noclprint data=smallmice;

class id cday;

model weight=day day*day / s;

random intercept /type=un subject=id v vcorr;

repeated cday / type=ar(1) subject=id   r;

ods output covparms=c17 fitstatistics=f17 dimensions=d17;

run;

data new17 (keep=descr value type model);

format descr $30.;

set c17(rename=(covparm=descr estimate=value)) f17 d17;

if (index(descr,'Res')>0 or index(descr, 'm')>0) and index(descr,'CC')=0 and index(descr,'mm')=0;;

type="AR(1) + RI";

model=17;

run;

proc sort data=new17;by descr;run;

*MACRO TO COMBINE inferences;
%macro nprme;

%do i = 1 %to 17;

 new&i

 %end;

 %mend nprme;

 data newtotal;

 format type $10.;

 set %nprme;

 by type;

 if first.type then count=1;

 else count+1;

run;

proc print data=newtotal;run;

proc freq data=newtotal;tables count;run;

proc sort data=newtotal;by model count;run;

proc transpose data=newtotal out=wide prefix=value;

by model type;

var value;

id count;

run;

data temp;

set wide;

parms=sum (of value4-value6);

label parms='# Parameters';

run;

proc sort data=temp;by model;run;

proc print data=temp label;

title 'Table';

var model type value6 value1 value2 value3 value7;

format value6 2.;

label
value1='REML: -2 Res Log Likelihood'
value2='AIC (smaller is better)'
value3='BIC (smaller is better)'
value4='Columns in X'
value5='Columns in Z'
value6='Covariance Parameters'
value7='Residual';

run;

5. Group= keyword.   

One last useful keyword “group=effect” for the repeated statement, where effect is a predictor in the class statement.  This gives different covariance parameters for the different groups in the ‘effect’ variable.  For example for the pain data:

TITLE1 'pediatric pain group keyword';

proc mixed method=reml covtest data=painlong;

class id cs tmt;

model l2paintol = cs tmt cs*tmt / s;

repeated / type=cs subject=id group=cs;

run;

has four covariance parameters, two for the attenders and two for the distracters.  

Similarly, 

repeated / type=un subject=id group=cs;

would give two fitted unstructured covariance models, one for attenders and one for distracters.  

